
G02 – Correlation and Regression Analysis

G02HAF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold
italicised terms and other implementation-dependent details.

1 Purpose

G02HAF performs bounded influence regression (M-estimates). Several standard methods are available.

2 Specification

SUBROUTINE G02HAF(INDW, IPSI, ISIGMA, INDC, N, M, X, IX, Y, CPSI,
1 H1, H2, H3, CUCV, DCHI, THETA, SIGMA, C, IC, RS,
2 WGT, TOL, MAXIT, NITMON, WORK, IFAIL)
INTEGER INDW, IPSI, ISIGMA, INDC, N, M, IX, IC, MAXIT,
1 NITMON, IFAIL
real X(IX,M), Y(N), CPSI, H1, H2, H3, CUCV, DCHI,
1 THETA(M), SIGMA, C(IC,M), RS(N), WGT(N), TOL,
2 WORK(4∗N+M∗(N+M))

3 Description

For the linear regression model
y = Xθ + ε

where y is a vector of length n of the dependent variable,

X is a n by m matrix of independent variables of column rank k,

θ is a vector of length m of unknown parameters,

and ε is a vector of length n of unknown errors with var(εi) = σ2:

G02HAF calculates the M-estimates given by the solution, θ̂, to the equation

n∑
i=1

ψ(ri/(σwi))wixij = 0 j = 1, 2, ...,m (1)

where ri is the ith residual i.e., the ith element of r = y −Xθ̂,

ψ is a suitable weight function,

wi are suitable weights,

and σ may be estimated at each iteration by the median absolute deviation of the residuals:

σ̂ = med
i
[|ri|]/β1

or as the solution to:
n∑

i=1

χ(ri/(σ̂wi))w
2
i = (n− k)β2

for suitable weight function χ, where β1 and β2 are constants, chosen so that the estimator of σ is
asymptotically unbiased if the errors, εi, have a Normal distribution. Alternatively σ may be held at a
constant value.

The above describes the Schweppe type regression. If the wi are assumed to equal 1 for all i then Huber
type regression is obtained. A third type, due to Mallows, replaces (1) by

n∑
i=1

ψ(ri/σ)wixij = 0, j = 1, 2, . . . ,m.

This may be obtained by use of the transformations

w∗
i ←

√
wi

y∗i ← yi

√
wi

x∗
ij ← xij

√
wi, j = 1, 2, . . . ,m

(see Marazzi Section 3).

For Huber and Schweppe type regressions, β1 is the 75th percentile of the standard Normal distribution.
For Mallows type regression β1 is the solution to

1
n

n∑
i=1

Φ(β1/
√
wi) = .75

where Φ is the standard Normal cumulative distribution function.

β2 is given by:

β2 =
∫ ∞

−∞
χ(z)φ(z) dz, in Huber case;

β2 =
1
n

n∑
i=1

wi

∫ ∞

−∞
χ(z)φ(z) dz, in Mallows case;

β2 =
1
n

n∑
i=1

w2
i

∫ ∞

−∞
χ(z/wi)φ(z) dz, in Schweppe case;

where φ is the standard Normal density, i.e.,

1√
2π

exp
(
−1
2
x2

)
.

The calculation of the estimates of θ can be formulated as an iterative weighted least-squares problem
with a diagonal weight matrix G given by

Gii =




ψ(ri/(σwi))
(ri/(σwi)),

ri �= 0

ψ′(0), ri = 0

where ψ′(t) is the derivative of ψ at the point t.

The value of θ at each iteration is given by the weighted least-squares regression of y on X . This is
carried out by first transforming the y and X by

ỹi = yi

√
Gii

x̃ij = xij

√
Gii, j = 1, 2, . . . ,m

and then using F04JGF. If X is of full column rank then an orthogonal-triangular (QR) decomposition
is used, if not, a singular value decomposition is used.

The following functions are available for ψ and χ in G02HAF.

(a) Unit Weights

ψ(t) = t, χ(t) = t2

2

this gives least-squares regression.

(b) Huber’s Function

ψ(t) = max(−c,min(c, t)), χ(t) =




t2

2
, |t| ≤ d

d2

2
, |t| > d

(c) Hampel’s Piecewise Linear Function

ψh1,h2,h3
(t) = −ψh1,h2,h3

(−t) =




t, 0 ≤ t ≤ h1

h1, h1 ≤ t ≤ h2

h1(h3 − t)/(h3 − h2), h2 ≤ t ≤ h3

0, h3 < t

χ(t) =




t2

2
, |t| ≤ d

d2

2
, |t| > d

(d) Andrew’s Sine Wave Function

ψ(t) =




sin t, −π ≤ π

0, |t| > π
χ(t) =




t2

2
, |t| ≤ d

d2

2
, |t| > d

(e) Tukey’s Bi-weight

ψ(t) =




t(1− t2)2, |t| ≤ 1

0, |t| > 1
χ(t) =




t2

2
, |t| ≤ d

d2

2
, |t| > d

where c, h1,h2,h3, and d are given constants.

Several schemes for calculating weights have been proposed, see Hampel et al. [1] and Marazzi [3]. As
the different independent variables may be measured on different scales, one group of proposed weights
aims to bound a standardized measure of influence. To obtain such weights the matrix A has to be found
such that:

1
n

n∑
i=1

u(‖zi‖2)ziz
T
i = I

and
zi = Axi

where xi is a vector of length m containing the ith row of X ,

A is a m by m lower triangular matrix,

and u is a suitable function.

The weights are then calculated as
wi = f(‖zi‖2)

for a suitable function f .

G02HAF finds A using the iterative procedure

Ak = (Sk + I)Ak−1

where Sk = (sjl),

sjl =
{
−min[max(hjl/n,−BL), BL] j > l
−min[max(1

2 (hjj/n− 1),−BD), BD] j = l

and

hjl =
n∑

i=1

u(‖zi‖2)zijzil

and BL and BD are bounds set at 0.9.

Two weights are available in G02HAF:

(i) Krasker–Welsch weights
u(t) = g1

(c

t

)

where g1(t) = t2 + (1 − t2)(2Φ(t)− 1)− 2tφ(t),
Φ(t) is the standard Normal cumulative distribution function,
φ(t) is the standard Normal probability density function,

and f(t) = 1
t .

These are for use with Schweppe type regression.
(ii) Maronna’s proposed weights

u(t) = c/t2 |t| > c
1 |t| ≤ c

f(t) =
√

u(t).

These are for use with Mallows type regression.

Finally the asymptotic variance-covariance matrix, C, of the estimates θ is calculated.

For Huber type regression
C = fH(X

TX)−1σ̂2

where
fH =

1
n−m

∑n
i=1 ψ

2 (ri/σ̂)(
1
n

∑n
i=1 ψ

′
(

ri

σ̂

))2κ
2

κ2 = 1 +
m

n

1
n

∑n
i=1

(
ψ′ (ri/σ̂)− 1

n

∑n
i=1 ψ′ (ri/σ̂)

)2

(
1
n

∑n
i=1 ψ

′
(

ri

σ̂

))2

See Huber [2] and Marazzi [4].

For Mallows and Schweppe type regressions C is of the form

σ̂

n

2

S−1
1 S2S

−1
1

where S1 =
1
nXTDX and S2 =

1
nXTPX .

D is a diagonal matrix such that the ith element approximates E(ψ′(ri/(σwi))) in the Schweppe case
and E(ψ′(ri/σ)wi) in the Mallows case.

P is a diagonal matrix such that the ith element approximates E(ψ2(ri/(σwi))w
2
i ) in the Schweppe case

and E(ψ2(ri/σ)w
2
i ) in the Mallows case.

Two approximations are available in G02HAF:

(1) Average over the ri

Schweppe Mallows

Di =


 1

n

n∑
j=1

ψ′
(

rj

σ̂wi

)
wi Di =


 1

n

n∑
j=1

ψ′
(rj

σ̂

)
wi

Pi =


 1

n

n∑
j=1

ψ2

(
rj

σ̂wi

)
w2

i Pi =


 1

n

n∑
j=1

ψ2
(rj

σ̂

)
w2

i

(2) Replace expected value by observed

Schweppe Mallows

Di = ψ′
(

ri

σ̂wi

)
wi Di = ψ′

(ri

σ̂

)
wi

Pi = ψ2

(
ri

σ̂wi

)
w2

i Pi = ψ2
(ri

σ̂

)
w2

i

See Hampel et al. [1] and Marazzi [4].

Note: There is no explicit provision in the routine for a constant term in the regression model. However,
the addition of a dummy variable whose value is 1.0 for all observations will produce a value of θ̂
corresponding to the usual constant term.

G02HAF is based on routines in ROBETH, see Marazzi [3].

4 References

[1] Hampel F R, Ronchetti E M, Rousseeuw P J and Stahel W A (1986) Robust Statistics. The
Approach Based on Influence Functions Wiley

[2] Huber P J (1981) Robust Statistics Wiley

[3] Marazzi A (1987) Weights for bounded influence regression in ROBETH Cah. Rech. Doc. IUMSP,
No. 3 ROB 3 Institut Universitaire de Médecine Sociale et Préventive, Lausanne

[4] Marazzi A (1987) Subroutines for robust and bounded influence regression in ROBETH Cah. Rech.
Doc. IUMSP, No. 3 ROB 2 Institut Universitaire de Médecine Sociale et Préventive, Lausanne

5 Parameters

1: INDW — INTEGER Input

On entry: specifies the type of regression to be performed.

If INDW = 0, Huber type regression.
If INDW < 0, Mallows type regression with Maronna’s proposed weights.
If INDW > 0, Schweppe type regression with Krasker–Welsch weights.

2: IPSI — INTEGER Input

On entry: specifies which ψ function is to be used.

If IPSI = 0, ψ(t) = t, i.e., least-squares.
If IPSI = 1, Huber’s function.
If IPSI = 2, Hampel’s piecewise linear function.
If IPSI = 3, Andrew’s sine wave.
If IPSI = 4, Tukey’s bi-weight.

Constraint: 0 ≤ IPSI ≤ 4.

3: ISIGMA — INTEGER Input

On entry: specifies how σ is to be estimated.

If ISIGMA < 0, σ is estimated by median absolute deviation of residuals.
If ISIGMA = 0, σ is held constant at its initial value.
If ISIGMA > 0, σ is estimated using the χ function.

4: INDC — INTEGER Input

On entry: if INDW �= 0, INDC specifies the approximations used in estimating the covariance
matrix of θ̂.

If INDC = 1, averaging over residuals.
If INDC �= 1, replacing expected by observed.
If If INDW = 0, INDC is not referenced.

5: N — INTEGER Input

On entry: the number n, of observations.

Constraint: N > 1.

6: M — INTEGER Input

On entry: the number m, of independent variables.

Constraint: 1 ≤ M < N.

7: X(IX,M) — real array Input/Output

On entry: the values of the X matrix, i.e., the independent variables. X(i, j) must contain the ijth
element of X , for i = 1, 2, . . . , n; j = 1, 2, . . . ,m.

If INDW < 0, then during calculations the elements of X will be transformed as described in Section
3. Before exit the inverse transformation will be applied. As a result there may be slight differences
between the input X and the output X.

On exit: unchanged, except as described above.

8: IX — INTEGER Input

On entry: the first dimension of the array X as declared in the (sub)program from which G02HAF
is called.

Constraint: IX ≥ N.

9: Y(N) — real array Input/Output

On entry: the data values of the dependent variable.

Y(i) must contain the value of y for the ith observation, for i = 1, 2, . . . , n.

If INDW < 0, then during calculations the elements of Y will be transformed as described in Section
3. Before exit the inverse transformation will be applied. As a result there may be slight differences
between the input Y and the output Y.

On exit: unchanged, except as described above.

10: CPSI — real Input

On entry: if IPSI = 1, CPSI must specify the parameter, c, of Huber’s ψ function. If IPSI �= 1 on
entry, CPSI is not referenced.

Constraint: if IPSI = 1, then CPSI > 0.0.

11: H1 — real Input
12: H2 — real Input
13: H3 — real Input

On entry: if IPSI = 2, H1, H2, and H3 must specify the parameters h1, h2, and h3, of Hampel’s
piecewise linear ψ function. H1, H2, and H3 are not referenced if IPSI �= 2.

Constraint: if IPSI = 2, then 0.0 ≤ H1 ≤ H2 ≤ H3 and H3 > 0.0.

14: CUCV — real Input

On entry:

If INDW < 0, CUCV must specify the value of the constant, c, of the function u for Maronna’s
proposed weights.
If INDW > 0, CUCV must specify the value of the function u for the Krasker–Welsch weights.
If INDW = 0, CUCV is not referenced.

Constraints:

if INDW < 0, CUCV ≥ M.
if INDW > 0, CUCV ≥

√
M.

15: DCHI — real Input

On entry: the constant, d, of the χ function. DCHI is not referenced if IPSI = 0, or if ISIGMA ≤ 0.

Constraint: if IPSI �= 0 and ISIGMA > 0, DCHI > 0.0.

16: THETA(M) — real array Input/Output

On entry: starting values of the parameter vector θ. These may be obtained from least-squares
regression. Alternatively if ISIGMA < 0 and SIGMA = 1 or if ISIGMA > 0 and SIGMA
approximately equals the standard deviation of the dependent variable, y, then THETA(i) = 0.0,
for i = 1, 2, . . . ,m may provide reasonable starting values.

On exit: THETA(i) contains the M-estimate of θi, for i = 1, 2, . . . ,m.

17: SIGMA — real Input/Output

On entry: a starting value for the estimation of σ. SIGMA should be approximately the standard
deviation of the residuals from the model evaluated at the value of θ given by THETA on entry.

Constraint: SIGMA > 0.0.

On exit: SIGMA contains the final estimate of σ if ISIGMA �= 0 or the value assigned on entry if
ISIGMA = 0.

18: C(IC,M) — real array Output

On exit: the diagonal elements of C contain the estimated asymptotic standard errors of the
estimates of θ, i.e., C(i, i) contains the estimated asymptotic standard error of the estimate contained
in THETA(i).

The elements above the diagonal contain the estimated asymptotic correlation between the estimates
of θ, i.e., C(i, j), 1 ≤ i < j ≤ m contains the asymptotic correlation between the estimates contained
in THETA(i) and THETA(j).

The elements below the diagonal contain the estimated asymptotic covariance between the estimates
of θ, i.e., C(i, j), 1 ≤ j < i ≤ m contains the estimated asymptotic covariance between the estimates
contained in THETA(i) and THETA(j).

19: IC — INTEGER Input

On entry: the first dimension of the array C as declared in the (sub)program from which G02HAF
is called.

Constraint: IC ≥M.

20: RS(N) — real array Output

On exit: the residuals from the model evaluated at final value of THETA i.e., RS contains the vector
(y −Xθ̂).

21: WGT(N) — real array Output

On exit: the vector of weights. WGT(i) contains the weight for the ith observation, for i =
1, 2, . . . , n.

22: TOL — real Input

On entry: the relative precision for the calculation of A (if INDW �= 0), the estimates of θ and the
estimate of σ (if ISIGMA �= 0). Convergence is assumed when the relative change in all elements
being considered is less than TOL.

If INDW < 0 and ISIGMA < 0, TOL is also used to determine the precision of β1.

It is advisable for TOL to be greater than 100*machine precision.

Constraint: TOL > 0.0.

[NP3390/19/pdf] G02HAF.7



G02HAF G02 – Correlation and Regression Analysis

23: MAXIT — INTEGER Input

On entry: the maximum number of iterations that should be used in the calculation of A (if INDW
�= 0), and of the estimates of θ and σ, and of β1 (if INDW < 0 and ISIGMA < 0).

A value of MAXIT = 50 should be adequate for most uses.

Constraint: MAXIT > 0.

24: NITMON — INTEGER Input

On entry: the amount of information that is printed on each iteration.

If NITMON = 0 no information is printed.
If NITMON �= 0 the current estimate of θ, the change in θ during the current iteration and
the current value of σ are printed on the first and every ABS(NITMON) iterations.

Also, if INDW �= 0 and NITMON > 0 then information on the iterations to calculate A is printed.
This is the current estimate of A and the maximum value of Sij (see Section 3).

When printing occurs the output is directed to the current advisory message unit (see X04ABF).

25: WORK(4∗N+M∗(N+M)) — real array Output

On exit: the following values are assigned to WORK:

WORK(1) = β1 if ISIGMA < 0, or WORK(1) = β2 if ISIGMA > 0.
WORK(2) = number of iterations used to calculate A.
WORK(3) = number of iterations used to calculate final estimates of θ and σ.
WORK(4) = k, the rank of the weighted least-squares equations.

The rest of the array is used as workspace.

26: IFAIL — INTEGER Input/Output

On entry: IFAIL must be set to 0, −1 or 1. Users who are unfamiliar with this parameter should
refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error or gives a warning (see Section 6).

For this routine, because the values of output parameters may be useful even if IFAIL �= 0 on exit,
users are recommended to set IFAIL to −1 before entry. It is then essential to test the value of
IFAIL on exit.

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Errors or warnings specified by the routine:

IFAIL = 1

On entry, N ≤ 1,

or M < 1,

or N ≤ M,

or IX < N,

or IC < M.

IFAIL = 2

On entry, IPSI < 0,

or IPSI > 4.

IFAIL = 3

On entry, SIGMA ≤ 0.0,

or IPSI = 1 and CPSI ≤ 0.0,

or IPSI = 2 and H1 < 0.0,

or IPSI = 2 and H1 > H2,

or IPSI = 2 and H2 > H3,

or IPSI = 2 and H1 = H2 = H3 = 0.0,

or IPSI �= 0 and ISIGMA > 0 and DCHI ≤ 0.0,

or INDW > 0 and CUCV <
√
M,

or INDW < 0 and CUCV < M.

IFAIL = 4

On entry, TOL ≤ 0.0,

or MAXIT ≤ 0.

IFAIL = 5

The number of iterations required to calculate the weights exceeds MAXIT. (Only if INDW �= 0).

IFAIL = 6

The number of iterations required to calculate β1 exceeds MAXIT. (Only if INDW< 0 and ISIGMA
< 0).

IFAIL = 7

Either the number of iterations required to calculate θ and σ exceeds MAXIT (NOTE, in this case
WK(3) = MAXIT on exit), or the iterations to solve the weighted least-squares equations failed
to converge. The latter is an unlikely error exit.

IFAIL = 8

The weighted least-squares equations are not of full rank.

IFAIL = 9

If INDW = 0 then (XTX) is almost singular.

If INDW �= 0 then S1 is singular or almost singular. This may be due to too many diagonal
elements of the matrix being zero, see Section 8.

IFAIL = 10

In calculating the correlation factor for the asymptotic variance-covariance matrix either the value
of

1
n

n∑
i=1

ψ′(ri/σ̂) = 0,

or
κ = 0,

or
n∑

i=1

ψ2(ri/σ̂) = 0.

See Section 8. In this case C is returned as XTX .

(Only if INDW = 0).

IFAIL = 11

The estimated variance for an element of θ ≤ 0.

In this case the diagonal element of C will contain the negative variance and the above diagonal
elements in the row and column corresponding to the element will be returned as zero.

This error may be caused by rounding errors or too many of the diagonal elements of P being zero.
See Section 8.

IFAIL = 12

The degrees of freedom for error, n− k ≤ 0 (this is an unlikely error exit), or the estimated value
of σ was 0 during an iteration.

7 Accuracy

The precision of the estimates is determined by TOL, see Section 5. As a more stable method is used
to calculate the estimates of θ than is used to calculate the covariance matrix, it is possible for the
least-squares equations to be of full rank but the (XTX) matrix to be too nearly singular to be inverted.

8 Further Comments

In cases when ISIGMA ≥ 0 it is important for the value of SIGMA to be of a reasonable magnitude.
Too small a value may cause too many of the winsorised residuals, i.e., ψ(ri/σ) to be zero or a value
of ψ′(ri/σ), used to estimate the asymptotic covariance matrix, to be zero. This can lead to errors
IFAIL = 8, 9 (if INDW �= 0), 10 (if INDW = 0) and 11.

G02HBF, G02HDF and G02HFF together carry out the same calculations as G02HAF but for user-
supplied functions for ψ, χ, ψ′ and u.

9 Example

The number of observations and the number of x variables are read in followed by the data. The option
parameters are then read in (in this case giving: Schweppe type regression with Hampel’s ψ function
and Huber’s χ function and then using the ‘replace expected by observed’ option in calculating the
covariances). Finally a set of values for the constants are read in.

After a call to G02HAF, θ̂, its standard error and σ̂ are printed. In addition the weight and residual for
each observation is printed.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details.
Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential
Introduction to this manual, the results produced may not be identical for all implementations.

* G02HAF Example Program Text
* Mark 14 Revised. NAG Copyright 1989.
* .. Parameters ..

INTEGER NIN, NOUT
PARAMETER (NIN=5,NOUT=6)
INTEGER NMAX, MMAX
PARAMETER (NMAX=8,MMAX=3)

* .. Local Scalars ..
real CPSI, CUCV, DCHI, H1, H2, H3, SIGMA, TOL
INTEGER I, IC, IFAIL, INDC, INDW, IPSI, ISIGMA, IX, J, M,

+ MAXIT, N, NITMON
* .. Local Arrays ..

real C(MMAX,MMAX), RS(NMAX), THETA(MMAX), WGT(NMAX),

G02HAF.10 [NP3390/19/pdf]



G02 – Correlation and Regression Analysis G02HAF

+ WORK(4*NMAX+MMAX*(NMAX+MMAX)), X(NMAX,MMAX),
+ Y(NMAX)

* .. External Subroutines ..
EXTERNAL G02HAF, X04ABF

* .. Executable Statements ..
WRITE (NOUT,*) ’G02HAF Example Program Results’

* Skip heading in data file
READ (NIN,*)
CALL X04ABF(1,NOUT)

* Read in number of observations and number of X variables
READ (NIN,*) N, M
WRITE (NOUT,*)
IF (N.GT.0 .AND. N.LE.NMAX .AND. M.GT.0 .AND. M.LE.MMAX) THEN

* Read in X and Y
DO 20 I = 1, N

READ (NIN,*) (X(I,J),J=1,M), Y(I)
20 CONTINUE

* Read in control parameters
READ (NIN,*) INDW, IPSI, ISIGMA

* Read in appropriate weight function parameters.
IF (INDW.NE.0) READ (NIN,*) CUCV, INDC
IF (IPSI.GT.0) THEN

IF (IPSI.EQ.1) READ (NIN,*) CPSI
IF (IPSI.EQ.2) READ (NIN,*) H1, H2, H3
IF (ISIGMA.GT.0) READ (NIN,*) DCHI

END IF
* Set values of remaining parameters

IX = NMAX
IC = MMAX
TOL = 0.5e-4
MAXIT = 50

* * Change NITMON to a positive value if monitoring information
* is required *

NITMON = 0
SIGMA = 1.0e0
DO 40 I = 1, M

THETA(I) = 0.0e0
40 CONTINUE

IFAIL = -1
*

CALL G02HAF(INDW,IPSI,ISIGMA,INDC,N,M,X,IX,Y,CPSI,H1,H2,H3,
+ CUCV,DCHI,THETA,SIGMA,C,IC,RS,WGT,TOL,MAXIT,NITMON,
+ WORK,IFAIL)

*
IF ((IFAIL.NE.0) .AND. (IFAIL.LT.7)) THEN

WRITE (NOUT,99999) ’G02HAF fails, IFAIL = ’, IFAIL
ELSE

IF (IFAIL.GE.7) THEN
WRITE (NOUT,99999) ’G02HAF returned IFAIL = ’, IFAIL
WRITE (NOUT,*)

+ ’ Some of the following results may be unreliable’
END IF
WRITE (NOUT,99998) ’Sigma = ’, SIGMA
WRITE (NOUT,*)
WRITE (NOUT,*) ’ THETA Standard’
WRITE (NOUT,*) ’ errors’
DO 60 J = 1, M

WRITE (NOUT,99997) THETA(J), C(J,J)

60 CONTINUE
WRITE (NOUT,*)
WRITE (NOUT,*) ’ Weights Residuals’
DO 80 I = 1, N

WRITE (NOUT,99997) WGT(I), RS(I)
80 CONTINUE

END IF
END IF
STOP

*
99999 FORMAT (1X,A,I2)
99998 FORMAT (1X,A,F10.4)
99997 FORMAT (1X,F12.4,F13.4)

END

9.2 Program Data

G02HAF Example Program Data

8 3 : N M

1. -1. -1. 2.1 : X1 X2 X3 Y
1. -1. 1. 3.6
1. 1. -1. 4.5
1. 1. 1. 6.1
1. -2. 0. 1.3
1. 0. -2. 1.9
1. 2. 0. 6.7
1. 0. 2. 5.5 : End of X1 X2 X3 and Y values

1 2 1 : INDW IPSI ISIGMA
3.0 0 : CUCV INDC (Only if INDW.ne.0)
1.5 3.0 4.5 : H1 H2 H3 (Only if IPSI.eq.2)
1.5 : DCHI (Only if IPSI.gt.0 .and. ISIGMA.gt.1)

9.3 Program Results

G02HAF Example Program Results

Sigma = 0.2026

THETA Standard
errors

4.0423 0.0384
1.3083 0.0272
0.7519 0.0311

Weights Residuals
0.5783 0.1179
0.5783 0.1141
0.5783 -0.0987
0.5783 -0.0026
0.4603 -0.1256
0.4603 -0.6385
0.4603 0.0410
0.4603 -0.0462

